From: Ralph at moondog@yahoo.com
There was some formatting stuff when I pasted it from Word.  
I hope I got rid of everything properly.

* PHONETICS ROUTINE

* 

*

*

PHONET   CSECT

         USING PHONET,12

         STM   14,12,12(13)

         LR    12,15

         L     6,0(1)                  GET LENGTH

* DE-COMMENT THIS LINE TO TEST IN NATURAL

*        L     6,0(6)                  **** TEMP FOR NAT TEST 

         L     7,4(1)                  GET ADDRESS OF WORD

         LA    8,CODE                  GET ADDRESS OF RESULT

* DE-COMMENT THIS LINE TO TEST IN NATURAL

*        L     8,8(1)                  **** TEMP FOR NAT TEST 

         ST    8,8(1)                  STORE IN EXPECTED LOCATION

         LA    9,20                    MAX OF 20 BYTES

         CR    9,6                     HOW MANY BYTES SENT

         BH    USELGTH                 IF 20 OR LESS USE REAL LENGTH

         LR    6,9                     ELSE USE 20

USELGTH  SR    5,5                     CLEAR A REG

         ST    5,CODET                 CLEAR TEMP CODE AREA

         ST    5,CODE                  AND FINAL AREA

         MVC   WORK,0(7)               MOVE WORD TO WORK

         OI    WORK,X'40'              SET FIRST BYTE TO UPPER CASE

         OI    WORK+1,X'40'            AND SECOND BYTE

         CLI   WORK,X'40'              IS FIRST BYTE BLANK?

         BE    NOMOVE                  YES - GO DIRECTLY TO TRANSLATE

*

* IF NAME IS NUMERIC USE UP TO FIVE POSITIONS AS PHONETIC KEY

*  - FIRST TWO POSITIONS WILL BE PLACED IN FIRST BYTE OF PHON KEY

*

         TM    WORK,X'F0'              IS FIRST POS NUMERIC?

         BNO   NONUMLP                 NO - DO SOUNDEX

         MVC   CODET(1),WORK           GRAB FIRST BYTE

         CH    6,=H'1'                 IS WORD MORE THAN ONE BYTE?

         BNH   NUMPK                   NO - DONT LOOK AT SECOND BYTE

         TM    WORK+1,X'F0'            IS SECOND BYTE NUMERIC?

         BNO   NUMPK                   NO - GO PACK IT

         MVC   CODET+1(1),WORK+1       GRAB SECOND BYTE

NUMPK    PACK  CODE(2),CODET(3)        ESTABLISH FIRST POSITION OF 

CODE

         MVI   CODE+1,X'00'            CLEAR TRAILING BYTE

         ST    5,CODET                 CLEAR TEMP FIELD

         CH    6,=H'2'                 MORE THAN 2 BYTES?

         BNH   LOOPEND                 NO - END OF LOOP

         LR    9,6                     GET LENGTH FOR LOOP CONTROL

         SH    9,=H'2'                 BACK OFF TWO BYTES

         TM    WORK+1,X'F0'            WAS SECOND BYTE NUMERIC?

         BNO   LOOPEND                 NO - DONT LOOP

         LA    3,CODET+1               POINT TO 2ND BYTE OF TEMP 

FIELD

         LA    4,CODET+3               AND LAST BYTE

         LA    5,WORK+2                POINT TO NEXT BYTE OF WORD

NUMLOOP  TM    0(5),X'F0'              IS THIS POSITION NUMERIC?

         BNO   LOOPEND                 NO - GO NO FURTHER   

         MVC   0(1,3),0(5)             YES - GRAB BYTE

         LA    5,1(5)                  BUMP TO NEXT BYTE OF WORD

         CR    3,4                     IS THIS THE END OF THE CODE?

         BE    LOOPEND                 YES - ALL DONE

         LA    3,1(3)                  BUMP TO NEXT POS OF TEMP CODE

         BCT   9,NUMLOOP               GO THROUGH ENTIRE WORD

         B     LOOPEND                 TIL THE END

*

* THE FOLLOWING LOOP IS USED TO TRANSLATE THE FIRST TWO

* POSITIONS OF THE WORD TO A PHONETIC VALUE - PREVIOUS TO

* RUNNING THE ENTIRE WORD THROUGH STANDARD SOUNDEX

*

NONUMLP  EQU   *

         LA    3,TWOSTBL               GET ADDR OF 2 LTR TABLE

         L     4,TWOSLGTH              AND NUMBER OF ENTRIES

         LA    9,WORK                  GET ADDRESS OF WORD

         CH    6,=H'1'                 IS LENGTH ONLY ONE?

         BE    DOONESLP                YES - DONT CHECK TWOS TABLE

TWOSLP   CLC   0(2,9),0(3)             CHECK TABLE FOR BYTES

         BNE   TWOSAGN                 NOT SAME - DONT REPLACE

         MVC   0(2,9),2(3)             MOVE IN REPLACEMENT LTRS

         B     NOTEXX                  DONE WITH TWOS TABLE         

TWOSAGN  EQU   *

         LA    3,4(3)                  BUMP TO NEXT ENTRY

         BCT   4,TWOSLP                GO THROUGH WHOLE TABLE

DOONESLP EQU   *                       NO ONES TABLE FOR NOW

         CLI   0(9),C'X'               JUST LOOK FOR X

         BNE   NOTEXX                  

         MVI   0(9),C'Z'               AND MAKE IT Z 

NOTEXX   EQU   *

*

* NOW DO STANDARD SOUNDEX ON RESULT

*

         MVC   CODE(1),WORK            MOVE FIRST BYTE TO CODE

NOMOVE   TR    WORK,TRTABLE            TRANSLATE LETTERS TO VALUES

* NOW WORK WITH RESULTS

         LR    9,6                     GET LENGTH OF WORD

         BCT   9,DOLOOP                ONE LESS FOR LOOP

         B     LOOPEND                 LENGTH WAS ONE - DONT LOOP

DOLOOP   LA    6,9                     SET TO BLANK VALUE

         LA    10,WORK+1               POINT TO SECOND BYTE

         IC    5,WORK                  GET VAL OF 1ST LETTER FOR LAST

         LA    4,CODET+1               POINT TO 2ND BYTE OF CODE

         LA    3,CODET+4               POINT TO LAST BYTE OF CODE

LOOP1    EQU   *

         SR    11,11                   CLEAR TEST REG

         IC    11,0(10)                GRAB NEXT BYTE OF WORD

         LTR   11,11                   WAS IT A ZERO VALUE LETTER?

         BZ    LOOP23                  YES - WE DONT WANT IT

         CR    11,6                    WAS IT A BLANK?

         BE    LOOPSM                  YES - IGNORE IT

         CR    5,11                    WAS IS SAME AS LAST VALUE?

         BE    LOOPSM                  YES - DONT WANT IT

         LR    5,11                    NOT SAME - SAVE AS LAST

         STC   11,0(4)                 AND PUT IT IN CODE   

         LA    4,1(4)                  BUMP CODE ADDRESS

         CR    3,4                     IS THIS END OF CODE FIELD?

         BNH   LOOPEND                 YES - FAR ENOUGH

         B     LOOPSM                  NO - DO THE NEXT ONE

*

LOOP23   EQU   *

         SR    5,5                     CHAR WAS ZERO VAL - CLEAR LAST

LOOPSM   LA    10,1(10)                BUMP TO NEXT LETTER

         BCT   9,LOOP1                 LOOP THROUGH ENTIRE WORD

* COMPRESS CODE TO 3 BYTES

LOOPEND  EQU   *

         PACK  CODE+1(2),CODET+1(3)    PICK UP NUMERIC VALUES

* DE-COMMENT THIS LINE TO TEST IN NATURAL

*        MVC   0(4,8),CODE             **** TEMP FOR NAT TEST 

*

* RETURN

RTN      EQU   *

         LM    14,12,12(13)            RESTORE REGS

         SR    15,15                   CLEAR RESPONSE

         BR    14                      GO TO CALLER

*

         LTORG

*

         CNOP  0,4                     GET A BOUNDRY

CODE     DS    CL4

CODET    DS    CL4

TWOSLGTH DC    F'23'                   INCREASE IF ADDING TO   

*                                      TWOSTBL

WORK     DS    CL20

*

* ADD HERE IF YOU FIND MORE - INCREASE TWOSLGTH ABOVE

*

TWOSTBL  DC    C'PHFAPFFAGFFAKNNAGNNAPNNATZZATSZAWRRAPSSACAKA'

         DC    C'MNNUCUKUCOKOTCCHCZZACSZASZZAYBIBYNINYVIVYPIP'

         DC    C'TSIS'

*       

TRTABLE  DS    0CL256

         DC    64X'00'

         DC    X'09'

               --A.B.C.D.E.F.G.H.I.    LOWERCASE

         DC    X'000102030001020000'

         DC    7X'00'

               --J.K.L.M.N.O.P.Q.R.    LOWERCASE

         DC    X'020204050500010206'

         DC    8X'00'

               --S.T.U.V.W.X.Y.Z.      LOWERCASE

         DC    X'0203000100020002'  

         DC    23X'00'

               --A.B.C.D.E.F.G.H.I.    UPPERCASE

         DC    X'000102030001020000'

         DC    7X'00'

               --J.K.L.M.N.O.P.Q.R.    UPPERCASE

         DC    X'020204050500010206'

         DC    8X'00'

               --S.T.U.V.W.X.Y.Z.      UPPERCASE

         DC    X'0203000100020002'  

         DC    22X'00'

*   

         END

Ralph W. Partlow

Adabas DBA 

Satyam Computer Services

Nissan North America, Nashville

**********************

** Note from Dieter **

**********************

Ralph,
 

Nearly forgot that we are using user exit 3 -- the phonetic routine -- since 1987. 
 

Sorry for the confusion.
 

Maybe I will describe phonetic descriptors and their usage on my Web site.
 

Dieter Storr, LA Times
 

 

* 
* SITE INSTALLED PHONETIC ROUTINE 
* 02/02/87 
* RALPH PARTLOW, DAVID GAHRING 
* LA TIMES 
* FIXED ON 5/9/89 BY BARRY LIPSKER 
ADAUX3  CSECT 
ADAUX3  AMODE 31 
ADAUX3  RMODE 24 
        USING ADAUX3,12 
        STM 14,12,12(13) 
        LR 12,15 
        L 6,0(1) GET A(LENGTH) 
<snip>
* 
* THE FOLLOWING TWO LOOPS ARE USED TO TRANSLATE THE FIRST ONE OR TWO 
* POSITIONS OF THE WORD TO A PHONETIC VALUE - PREVIOUS TO RUNNING THE 
* ENTIRE WORD THROUGH THE STANDARD 'SOUNDEX' ALOGRITHM. THE TWO 
* CHARACTER TABLE IS CHECKED FIRST AS IT MIGHT OVERRIDE AN EFFECT 
* OF THE ONE CHARACTER CHECK. 
<snip>
*
WORK    DS CL20                                            
ONESTBL DC C'XZ'                                           
TWOSTBL DC C'PHFAPFFAGFFAKNNAGNNAPNNATZZATSZAWRRAPSSACAKA' 
        DC C'MNNUCUKUCOKO'                                 
TRTABLE DS 0CL256                                          
        DC 64X'00'                                         
        DC X'09'                                           
        DC 64X'00'                                         
*          - A.B.C.D.E.F.G.H.I.     LOWERCASE              
        DC X'000102030001020000'                           
        DC 7X'00'                                          
*          - J.K.L.M.N.O.P.Q.R.     LOWERCASE              
        DC X'020204050500010206'                           
        DC 8X'00'                                          
*          - S.T.U.V.W.X.Y.Z.       LOWERCASE              
        DC X'0203000100020002'                             
        DC 23X'00'                                         
*          - A.B.C.D.E.F.G.H.I.     UPPERCASE              
        DC X'000102030001020000'                           
        DC 7X'00'                                  
*          - J.K.L.M.N.O.P.Q.R.     UPPERCASE      
        DC X'020204050500010206'                   
        DC 8X'00'                                  
*          - S.T.U.V.W.X.Y.Z.       UPPERCASE      
        DC X'0203000100020002'                     
        DC 22X'00'                                 
*                                                  
PATCH   DC 200X'00'                                
        END
